/*
 * Copyright 2022 The Android Open Source Project
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *      https://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

plugins {
    id 'com.android.application' version '8.6.0' apply false
    id 'com.android.library' version '8.6.0' apply false
    id 'org.jetbrains.kotlin.android' version '2.0.0' apply false
    id 'com.diffplug.spotless'  version '6.9.0' apply true
}

subprojects {
    spotless {
        java {
            target '**/*.java'
            googleJavaFormat('1.8').aosp()
            licenseHeaderFile project.rootProject.file('scripts/copyright')
        }
        kotlin {
            target '**/*.kt'
            ktlint('0.47.0').userData(['android': 'true'])
            licenseHeaderFile project.rootProject.file('scripts/copyright')
        }
        groovyGradle {
            target '**/*.gradle' // default target of groovyGradle
            // the Groovy Eclipse formatter extends the Java Eclipse formatter,
            // so it formats Java files by default (unless `excludeJava` is used).
            greclipse().configFile(project.rootProject.file('scripts/greclipse.properties'))
            licenseHeaderFile project.rootProject.file('scripts/copyright'), '(plugins|pluginManagement)'
        }
    }
}